
Hallo zusammen,

ich musste feststellen, dass ich eins nicht verstehe, obwohl es wirklich um "Basisc" geht.
Also, die Frage:
warum produzieren zwei folgende Data Steps verschiedene Dateien?

data test2;
set test;
set test;
set test;
run;

data test3;
do i=1 to 3;
set test;
end;
run;

(Was die Datei "test" ist, spielt keine Rolle, also z.B. so eine:

data test;
input v1 $ v2;
datalines;
aa 1
bb 3
cc 4
;

)

Irgendwie dachte ich naiv, dass
do i=1 to 3;
set test;
end;
einfach dreimal "set test" ist, aber es ist offensichtlich nicht so.
Warum aber?

Danke im Voraus.

Einfache Frage zu Data Step

15 November, 2012 - 10:29 — HansKneilmannSchwere Frage

15 November, 2012 - 10:38 — HansKneilmann

15 November, 2012 - 18:22 — JanHeuer

eine schwere Frage, da müßte ich in die alten Kurs-Unterlagen Programmoptimierung (bei mir hieß
der Kurs immer nur Performanceoptimierung) eintauchen. Ich versuche ein Erklärung ohne.

Der Data Step beginnt oben (klar) und gibt unten (beim run) den aktuellen Datensatz aus, es sei

denn man verwendet den output-Befehl.

Deshalb hat test3 nur eine observation, da der Data Step nur einmal bis zum unteren Ende

durchlaufen wird. Das er unterwegs durch die Schleife gejagt wird zählt nicht, es verändert nur den
Inhalt der Variablen.

Warum test2 3 observations enthält kann ich nicht sagen. Der Data Step ohne den expliziten

output-Befehl hat ein implizites output mit wahrscheinlich ganz eigenen Regeln, die ich nicht

kenne. Ich denke bei diesen Regeln liegt die Erklärung.

Warum hat test2 drei verschiedene observations? Ganz klar: The number of obs in the new data
set is the number of obs in the smallest original data set.
Die Inhalte überschreiben sich jeweils gegenseitig: If the data sets contain common variables, the
values read in from the last data set replace those read in from earlier ones.
Beide Zitate aus SAS Language, Reference Guide, Version 6 (Papier Version aus den Anfangsjahren)
Rules of the SAS Language, Combinig SAS Data Sets, One-to-One Reading.
Dort steht noch ein wichtiger Satz: Using multiple SET statements to combine obs can produce
undesirable results.
Toller Satz, oder?
Und weiter geht's: Test your program on representative samples of the data sets before using this
method to combine them.
Dem ist nichts hinzuzufügen.

Gruß
Hans Kneilmann, Schäfer Shop GmbH (SSI)

Edit: Den letzten Absatz ab "Warum hat test2 drei" habe ich um 10:28 komplett neu geschrieben,
die erste Version war hinfällig

Link zur Doku
hier noch die passende Doku zu
One-to-One Reading
Am besten runter rollen bis zur Überschrift
One-to-One Reading and One-to-One Merging

seltsames Verhalten
Hallo Hans, Hallo Dr. Boogie,

Bei test2 wird in diesem Beispiel 19 mal 3 mal dieselbe Zeile eingelesen. danach erfolgt eine
Ausgabe. Ein Data-Step ist eine implizite Schleife.



/* Beispieldaten */
Data test;id = _N_; set sashelp.class ;run;
 
data test2;
  set test;
  set test;
  set test;
run;
/*NOTE: There were 19 observations read from the data set WORK.TEST.*/
/*NOTE: There were 19 observations read from the data set WORK.TEST.*/
/*NOTE: There were 19 observations read from the data set WORK.TEST.*/
/*NOTE: The data set WORK.TEST2 has 19 observations and 6 variables.*/
/*NOTE: DATA statement used (Total process time):*/
/*      real time           0.01 seconds*/
/*      cpu time            0.01 seconds*/
/*
Es sieht so aus, als ob SAS in diesem Beispiel folgenden Ablauf hat.
  set test (firstobs=1 obs=1); 
  set test (firstobs=1 obs=1); 
  set test (firstobs=1 obs=1);
  output; 
  set test (firstobs=2 obs=2); 
  set test (firstobs=2 obs=2); 
  set test (firstobs=2 obs=2);
  output;
  ...
  set test (firstobs=n obs=n); 
  set test (firstobs=n obs=n); 
  set test (firstobs=n obs=n);
  output; 
*/

Bei Test3 wird der Zähler mit jedem Set-Befehl hochgesetzt, da eine explizite Schleife. In jedem
Schleifendurchgang wird genau ein Datensatz gelesen. Nach dem dritten mal erfolgt eine
Ausgabe.



data test3;
  do i=1 to 3;
    set test;
  end;
  drop i;
run;
/*NOTE: There were 19 observations read from the data set WORK.TEST.*/
/*NOTE: The data set WORK.TEST3 has 6 observations and 6 variables.*/
/*NOTE: DATA statement used (Total process time):*/
/*      real time           0.01 seconds*/
/*      cpu time            0.01 seconds*/
Proc print;run;
/*    Obs    id    Name       Sex    Age    Height    Weight*/
/**/
/*    1      3    Barbara     F      13     65.3        98*/
/*    2      6    James       M      12     57.3        83*/
/*    3      9    Jeffrey     M      13     62.5        84*/
/*    4     12    Judy        F      14     64.3        90*/
/*    5     15    Philip      M      16     72.0       150*/
/*    6     18    Thomas      M      11     57.5        85*/
/*
6 ist 19\3 Rest 1
 
hier ist der Ablauf folgender
 
  set test (firstobs = 1 obs = 1);
  set test (firstobs = 2 obs = 2);
  set test (firstobs = 3 obs = 3);
  output;
  set test (firstobs = 4 obs = 4);
  set test (firstobs = 5 obs = 5);
  set test (firstobs = 6 obs = 6);
  output;
  ...
  set test (firstobs = n       obs = n);
  set test (firstobs = n + 1 obs = n +1 );
  set test (firstobs = n + 2 obs = n +2 ); /* hier erfolgt kein Output */
  output;
*/

Um z. B. eine SQL-ähnliche Ungleichheitsbeziehung zu erstellen, braucht man die Unabhängikeit
des Befehls in der Do-Schleife. Ohne die Point-Option geht es da aber auch nicht.

Data class;
  set sashelp.class;
  Alter = Age;
  Name_aelter = name;
  do i = 1 to  anz;
    set sashelp.class  point=i nobs=anz;
    if Alter > Age then output;
  end;
run;
NOTE: There were 19 observations read from the data set SASHELP.CLASS.
NOTE: The data set WORK.CLASS has 145 observations and 7 variables.
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.00 seconds



ohne Point beinflussen sich die Set-Befehle gegenseitig, weil in der Schleife der Zähler
hochgezählt wird, der erste Set-Befehl zählt den Zähler aber erst am Ende desDurchgangs hoch.

Data class;
  set test;
  do i = 1 to  3;
    set test;
  end;
run;
NOTE: There were 7 observations read from the data set WORK.TEST.
NOTE: There were 19 observations read from the data set WORK.TEST.
NOTE: The data set WORK.CLASS has 6 observations and 7 variables.
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.00 seconds

Schöne Grüße

Jan

